COMP 410
Summer 2024
Midterm Practice Exam #2 (Solutions)

This is representative of the kinds of topics and kind of questions you may be asked on the midterm. This practice exam, along with assignments 3-6 and the in-class handouts from week 3 and beyond, are intended to be comprehensive of everything on the exam. That is, I will try not to ask anything that's not somehow covered by those sources.

I will announce the exact cutoff for the handouts on Wednesday. While I won't intentionally ask questions from exam #1 material, some material naturally builds on earlier material (e.g., unification).

You are permitted to bring two 8.5 x 11 sheets of paper into the exam with you, as long as they have handwritten notes on them. Both sides of both sheets can be used. To be clear, these must be entirely handwritten.

More Recursion

1.) Write a procedure named `allEqual` which will succeed if all list elements are equal to each other according to unification (=). You may introduce any helpers you wish. Example calls are below:

```
?- allEqual([]).
true.
?- allEqual([1, 1, 1]).
true.
?- allEqual([1, 2, 3]).
false.
?- allEqual([1, X, 1]).
X = 1.
?- allEqual([A, B]).
A = B.
?- allEqual([X, 1, 2]).
false.
```

```
allEqual([]).
allEqual([_]).
allEqual([H, H|Rest]) :-
    allEqual([H|Rest]).
```

2.) Write a procedure named `zip`, which takes two lists of the same length, an output list of the same length. The output list is a list of `pair` structures, where each `pair` holds an element from each list, preserving order. If the lists are not the same length, `zip` should fail, though you shouldn't need to explicitly check the length. Example calls are below:

```
?- zip([], [], Output).
Output = [].
?- zip([hello], [goodbye], Output).
Output = [pair(hello, goodbye)].
?- zip([1, 2, 3], [a, b, c], Output).
Output = [pair(1, a), pair(2, b), pair(3, c)].
?- zip([A, B], [C, D], Output).
Output = [pair(A, C), pair(B, D)].
?- zip([foo], [bar, baz], Output).
false.
?- zip([foo, bar], [baz], Output).
false.
```

```
zip([], [], []).
zip([H1|T1], [H2|T2], [pair(H1, H2)|Rest]) :-
    zip(T1, T2, Rest).
```

3.) We can represent a binary tree in Prolog using the following:

- `leaf`: At atom representing a leaf node
- `internal(tree, value, tree)`: A structure recursively containing two trees and some integer value

Write a procedure named sumTree that will compute the sum of all the elements in a given binary tree. Leaf nodes are defined to have a sum of 0. Example queries are shown below:

```
?- sumTree(leaf, Sum).
Sum = 0.
?- sumTree(internal(leaf, 5, leaf), Sum).
Sum = 5.
?- sumTree(internal(internal(leaf, 1, leaf),
                    2,
                    internal(leaf, 3, leaf)),
           Sum).
Sum = 6.
```

```
sumTree(leaf, 0).
sumTree(internal(Left, Value, Right), Sum) :-
  sumTree(Left, LeftSum),
  sumTree(Right, RightSum),
  Sum is LeftSum + Value + RightSum.
```

4.) Consider the following code, computing the length of a list:

```
len([], 0).
len([_|T], Len) :-
     len(T, TLen),
     Len is TLen + 1.
```

4.a) This procedure is not very efficient when it comes to memory. Why is it inefficient?

It uses O(N) stack space since it is not tail-recursive.

4.b) Rewrite this procedure to be more efficient with memory. You may introduce a helper procedure if desired.

```
len(List, Len) :-
    len(List, 0, Len).

len([], Accum, Accum).
len([_|T], Accum, Len) :-
    NewAccum is Accum + 1,
    len(T, NewAccum, Len).
```

5.) Consider the following code which appends two lists together:

```
append([], List, List).
append([H|T], List, Result) :-
  append(T, List, Rest),
  Result = [H|Rest].
```

5.a) This procedure is not very efficient when it comes to memory. Why is it inefficient?

It uses O(N) stack space since it is not tail-recursive.

5.b) Rewrite this procedure to be more efficient with memory. You may introduce a helper procedure if desired.

```
append([], List, List).
append([H|T], List, [H|Rest]) :-
  append(T, List, Rest).
```

6.) Define a procedure named `isPrime` which will determine if a given input number is prime. You may introduce any helpers you wish. Example queries follow:

```
?- isPrime(2).
true .
?- isPrime(3).
true .
?- isPrime(4).
false.
```

As a hint, the following Java-like code:

```
int x = y % z;
```

...is equivalent to the following Prolog code:

```
X is mod(Y, Z)
```

```
isPrime(Num) :-
    FirstTest is Num - 1,
    isPrime(Num, FirstTest).

% isPrime: Number, CurrentTest
isPrime(_, 1).
isPrime(Num, Test) :-
    Test > 1,
    NonZero is mod(Num, Test),
    NonZero \== 0,
    NewTest is Test - 1,
    isPrime(Num, NewTest).
```

Test Case Generation

7.) Consider the following grammar-based definition of simplistic SQL queries:

$$c \in ColumnName \quad t \in TableName$$
$$q \in SQLQuery ::= \texttt{select}\ c\ \texttt{from}\ t;$$

7.a) Assume the only possible columns are named `c1` and `c2`, and the only possible tables are named `t1` and `t2`. Write a generator of valid SQL query ASTs. An example of a valid AST is `select(c1, t1)`. Do not simply hardcode all possible ASTs.

```
columnName(c1).
columnName(c2).

tableName(t1).
tableName(t2).

sql(select(C, T)) :-
    columnName(C),
    tableName(T).
```

7.b) Bounds or related mechanisms are not necessary for this problem, at least as described. Why?

From the description, there are a reasonably finite number of possible ASTs. (Another possible answer) from the implementation, there is no recursion, which would potentially allow us to "spam" the same rule indefinitely.

7.c) Name a change to this problem which would necessitate adding a bound or a related mechanism, and explain why such a change would add this necessity.

Add a column or table name generator. Another answer is to add support for `while` clauses, which can be chained arbitrarily long with Boolean operators like AND. In both cases, these features make the space infinite, requiring us to inject failure somewhere to prevent us from producing repetitive-looking ASTs.

8.) Consider the following grammar:

```
tree ::= `node(` tree `,` tree `)` | `leaf`
```

8.a.) Write a generator for `tree` below. It's ok if the generator gets "stuck" generating similar values over and over again.

```
gen(leaf).
gen(node(Left, Right)) :-
    gen(Left),
    gen(Right).
```

8.b.) Write a modified version of your prior generator, which takes an additional depth bound, and will only generate values that are no deeper than this bound.

```
genBound(_, leaf).
genBound(Bound, node(Left, Right)) :-
    Bound > 0,
    NewBound is Bound - 1,
    genBound(NewBound, Left),
    genBound(NewBound, Right).
```

Logic Programming in Python (Depends on Wednesday)

9.) Consider the following Prolog procedure:

```
isName(alice).
isName(bob).
isName(janet).
isName(bill).
```

Write an equivalent generator function in Python, named `isName`. Each name should be represented as a string. As a hint, `isName` should not take any parameters.

```
def isName():
  yield "alice"
  yield "bob"
  yield "janet"
  yield "bill"
```

10.) Consider the following Prolog procedure:

```
naturalNumber(0).
naturalNumber(N) :-
  naturalNumber(NMinusOne),
  N is NMinusOne + 1.
```

Write an equivalent generator function in Python, named `naturalNumber`. As a hint, `naturalNumber` should not take any parameters.

```
def naturalNumber():
  yield 0
  for nMinusOne in naturalNumber():
    yield nMinusOne + 1
```